

Radiohiiliajoitus

Jokaiselle radioaktiiviselle atomiytimelle on olemassa tietty todennäköisyys sen hajoamiseen määrätyllä aikavälillä. Hajoavien atomien lukumäärä ΔN on siten suoraan verrannollinen aktiivisten atomiytimien lukumäärään N ja lyhyellä aikavälillä myös likimain tarkasteltavaan aikaan Δt :

$$\Delta N \approx -\lambda N \Delta t.$$

Tämä voidaan kirjoittaa muotoon

$$\frac{\Delta N}{\Delta t} \approx -\lambda N.$$

Kun $\Delta t \rightarrow 0$, approksimatiivinen yhtälö muuttuu tarkaksi yhtälöksi ja vasen puoli lähestyy derivaattaa:

$$\frac{\partial}{\partial t} N = -\lambda N,$$

On siis saatu radioaktiivisen hajoamisen differentiaaliyhtälö. Tässä λ on verrannollisuuskerroin, ns. *hajoamisvakio*. Luonnollinen alkuehto on aineen määrä tarkastelun alussa hetkellä $t = 0$: $N(0) = N_0$.

Poistetaan mahdolliset aiemmat muuttujat ja syötetään differentiaaliyhtälö alkuehtoineen:

```
[ > restart;  
[ > yhtalo:= diff(n(t), t)=-lambda*n(t);  
                                yhtalo :=  $\frac{d}{dt} n(t) = -\lambda n(t)$   
[ > alkuehto:= n(0)=n[0];  
                                alkuehto :=  $n(0) = n_0$ 
```

Ratkaisuksi saadaan

```
[ > rtk:= dsolve({yhtalo, alkuehto}, n(t));  
                                rtk :=  $n(t) = n_0 e^{(-\lambda t)}$   
[ > ainemaara:= subs(rtk, n(t));  
                                ainemaara :=  $n_0 e^{(-\lambda t)}$ 
```

Radioaktiivisen aineen puoliintumisaika T_p on se aika, jossa radioaktiivisen aineen määrä on

vähentynyt puoleen. Hajoamisvakio voidaan lausua puoliintumisajan avulla:

```
> hajvakio:= solve(subs(t=T[p],ainemaara)=n[0]/2, lambda);
```

$$hajvakio := \frac{\ln(2)}{T_p}$$

Tällöin ainemäärä saa muodon

```
> subs(lambda=hajvakio, ainemaara):  
ainemaaratoisin:= simplify(%);
```

$$ainemaaratoisin := n_0 2^{\left(-\frac{t}{T_p}\right)}$$

Jokainen puoliintumisajan pituinen ajanjakso merkitsee ainemäärän puolittumista:

```
> seq(subs(t=k*T[p], ainemaaratoisin), k=0..10);
```

$$n_0, \frac{1}{2} n_0, \frac{1}{4} n_0, \frac{1}{8} n_0, \frac{1}{16} n_0, \frac{1}{32} n_0, \frac{1}{64} n_0, \frac{1}{128} n_0, \frac{1}{256} n_0, \frac{1}{512} n_0, \frac{1}{1024} n_0$$

Esimerkkinä olkoon eräessä fossiilinäytteessä havaittu radiohiilipitoisuus, joka oli 9 % vapaana olevan aineen radiohiilipitoisuudesta. Radiohiilen puoliintumisajaksi tiedetään 5730 vuotta. Näytteen ikä voidaan tämän perusteella määrittää, jos oletetaan, että luonnossa vapaana olevan aineen radiohiilipitoisuus ei ole muuttunut eikä näytteessä ole sen maahan joutumisen jälkeen tapahtunut hiiliaineenvaihduntaa. Näytteeseen aikoinaan sitoutunut radiohiili on tällöin aktiivisuutensa takia vähitellen vähentynyt todetulle yhdeksän prosentin tasolle. Saadaan siis ehto

```
> ehto:= 0.09*n[0]=subs(T[p]=5730, ainemaaratoisin);
```

$$ehto := 0.09 n_0 = n_0 2^{\left(-\frac{t}{5730}\right)}$$

Näytteen iäksi saadaan

```
> solve(ehto, t);
```

$$19905.62571$$

```
>
```

siis lähes 20000 vuotta.

Tehtävä

Plutoniumin isotoopin 239 puoliintumisaika on noin 24100 vuotta. Kauanko kestää, ennen kuin

plutoniumia sisältävän materiaalin aktiivisuus on laskennut a) 10 prosenttiin, b) yhteen prosenttiin alkuperäisestä? Voidaanko määritellä 'kymmenesosaan laskemisaika' samaan tapaan kuin puoliintumisaika? Millainen on näiden aikojen suhde?

Linkit

[radioaktiivinen hajoamisketju](#) (radahaj2.mws)

SKK & MS 12.07.2001